┌────────────────┐

│VII - Procedures│

└────────────────┘

VII.1 - Definiçäo

        ---------

        Uma das técnicas mais utilizadas e tida como vantajosa na confecçäo

        de programas grandes é  a  modularizaçäo.  Consiste  em  dividir  o

        programa   em  diversos  módulos  ou  subprogramas,  de certa forma

        independentes uns dos outros. Existe um módulo que é o principal, a

        partir do qual säo chamados os outros módulos, esse módulo recebe o

        nome de programa principal, enquanto que os outros säo chamados  de

        subprogramas. No  sistema  Turbo Pascal, existem dois tipos de sub-

        programas, a saber:

                - Procedures (procedimentos)

                - Functions (funçöes)

        A  procedure é   como  se  fosse  um  programa. Ela tem a estrutura

        praticamente  igual  a de um programa, como veremos mais adiante. A

        procedure deve ser ativada(chamada) pelo programa principal ou  por

        uma outra procedure, ou até por ela mesma.

VII.2 - Declaraçäo de procedures

        ------------------------

        Uma procedure tem praticamente a mesma estrutura de um programa, ou

        seja, ela contém um cabeçalho, área de declaraçÆes e  o  corpo   da

        procedure.  Na área   de declaraçöes, podemos ter as seguintes sub-

        áreas:

           Label - Const - Type - Var - Procedures - Functions.

        Devemos salientar que tudo que for declarado dentro das subáreas só

        será  reconhecido dentro da procedure. Mais para frente, voltaremos

        a falar sobre isso.

        Exemplo:

        ---------------------------------------------------------

        Program Exemplo_1;  (* cabeçalho do programa *)

        Uses CRT;

        Procedure linha;      (* cabeçalho da procedure linha *)

        Var i : integer;      (* subárea Var da procedure linha *)

        Begin                 (* corpo da procedure linha *)

           for i:=1 to 80 do write('-');

        End;

        Begin                 (* corpo do programa principal *)

           ClrScr;

           linha;             (* ativaçäo da procedure linha *)

           writeln('teste');

           linha;             (* ativaçäo da procedure linha,

                                 novamente *)

        End.

        ---------------------------------------------------------

        O programa acima, pura e simplesmente faz o seguin-te:

        1-) Apaga a tela e coloca o cursor em 1,1

        2-) Ativa a procedure linha

        3-) Escreve a palavra teste

        4-) Ativa novamente a procedure linha.

        Por sua vez, a procedure linha traça uma linha a partir da  posiçäo

        atual do cursor. Uma observaçäo importantíssima a ser  feita  neste

        instante, é que a variável inteira i,  definida dentro da procedure

        linha só existe dentro da procedure,isto significa que toda vez que

        ativamos a procedure linha, a variável i é criada  e  toda  vez que

        saimos da procedure linha, ela é destruída.

VII.3 - Passagem de parâmetros.

        -----------------------

        No exemplo acima, ao ativarmos a procedure linha,näo houve passagem

        de parâmetros, mas poderia haver, repare no exemplo abaixo:

        Exemplo:

        ---------------------------------------------------------

        Program Exemplo;

        Uses CRT;

        Var i,j:integer;

        Procedure soma(x,y:integer);

        Begin

           writeln(x+y);

        end;

        Begin

           ClrScr;

           soma(3,4);

           i:=45;

           j:=34;

           soma(i,j);

        end.

        -------------------------------------------------------------------

        Como podemos reparar, a procedure soma depende de dois   parâmetros

        inteiros, e ao ativarmos esta procedure, devemos fornecer  os  dois

        parâmetros.Esses parâmetros podem ser dois números inteiros ou duas

        variáveis inteiras, obviamente deve haver compatibilidade entre  os

        parâmetros passados. Podemos também  passar  parâmetros  de   tipos

        diferentes, senäo vejamos:

        -------------------------------------------------------------------

        Program Exemplo_1;

        Uses CRT;

        Var i,j:integer;

        Procedure soma(x,y:integer;h,g:real);

        Begin

           writeln(x+y);

           writeln(h/g:10:2);

        end;

        Begin

           ClrScr;

           i:=34;

           j:=35;

           soma(i,j,3.4,4.5);

        End.

        -------------------------------------------------------------------

        Nos exemplos acima,houve passagem de parâmetros para as procedures,

        mas   elas  também  podem   passar  dados  de volta para o programa

        chamador, exemplo:

        -------------------------------------------------------------------

        Program exemplo;

        Uses CRT;

        Var i : Integer;

        Procedure Soma(x,y:Integer;Var z:Integer);

        Begin

           z:=x+y;

        End;

        Begin

           ClrScr;

           Soma(3,4,i);

           Writeln(i);

        End.

        -------------------------------------------------------------------

        Da forma como foi declarada a procedure soma, quando a ativamos com

        a sequência Soma(3,4,i), ocorrem as seguintes passagens:

        - O número 3 é passado para x

        - O número 4 é passado para y

        - O parâmetro z é passado para i.

        Como  podemos ver, houve passagem de dados do programa chamador para

        a procedure e da procedure para o programa chamador.

VII.4 - A declaraçäo forward.

        ---------------------

        Suponha o programa abaixo:

        ---------------------------------------------------------

        Program exemplo;

        Uses CRT;

        Procedure Soma(x,y:Integer);

        Begin

           linha;

           Writeln(x+y);

        End;

        Procedure Linha;

        Var i:integer;

        Begin

           For i:=1 to 80 do Write('-');

        End;

        Begin

           ClrScr;

           Soma(3,4);

        End.

        ---------------------------------------------------------

        Repare  que a procedure Soma chama uma procedure chamada linha.  No

        entanto, a procedure linha está declarada mais à frente e portanto,

        ao  compilarmos  o  programa,  o  compilador irá "reclamar" que näo

        conhece  o  identificador   Linha  e com justa razäo, isto porque a

        compilaçäo é feita de cima para baixo e da esquerda para a direita.

        Para  tanto, podemos usar a declaraçäo Forward, cuja finalidade é a

        de   indicar  ao compilador que determinada procedure está definida

        mais para frente.

        Exemplo:

        ---------------------------------------------------------

        Program exemplo;

        Uses CRT;

        Procedure Linha; Forward;

        Procedure Soma(x,y:Integer);

        Begin

           linha;

           Writeln(x+y);

        End;

        Procedure Linha;

        Var i:integer;

        Begin

           For i:=1 to 80 do Write('-');

        End;

        Begin

           ClrScr;

           Soma(3,4);

        End.

        ---------------------------------------------------------

        Agora sim, podemos compilar o programa sem erro.

VII.5 - O escopo de objetos num programa

        ---------------------------------

        Reparem o Exemplo abaixo:

        ---------------------------------------------------------

        Program Exemplo;

        Uses CRT;

        Const a=100;                     (* constante global *)

        Label fim;                       (* Label global *)

        Var i,x,y : Integer;             (* variáveis globais *)

        Procedure Linha;

        Var i : Integer;                 (* i é local à procedure

                                            linha *)

        Begin

           For i:=1 to 80 do Write('-');

        End;

        Procedure Teste;

        Procedure Sub_teste;              (* a procedure

                                           Sub_teste é local

                                           à procedure Teste *)

        Begin

           Write('Estive em sub_teste');

        End;

        Begin

           Sub_teste;

           Writeln;

        End;

        Begin

           ClrScr;

           i:=100;

           Linha;

           x:=20;

           y:=30;

           Teste;

           Linha;

           Writeln('i=',i,' y=',y,' x=',x);

        End.

        ---------------------------------------------------------

        Todos  os  elementos (constantes, variáveis, labels etc.) que forem

        definidos antes de começar o corpo do  programa,  säo  considerados

        globais e podem ser utilizados por todas as procedures, functions e

        o próprio programa. O espaço para tais elementos é criado durante a

        compilaçäo. Já, os elementos declarados dentro de uma procedure, só

        existem  dentro  da procedure, exemplo: ao declararmos uma variável

        dentro  de  uma  procedure, toda vez que ativarmos a procedure, tal

        variável  será criada e ao sairmos da procedure ela será destruída.

        Portanto, dizemos que esta variável é local à procedure.

        No  entanto,  se  repararmos bem no exemplo, veremos que existe uma

        variável i inteira declarada antes do início do  programa, portanto

        global,  e  outra  dentro da procedure linha, portanto local a esta

        procedure.  Mas näo há problema, pois o Turbo Pascal irá considerá-

        las   diferentes.  Quando  estivermos  dentro  do programa, teremos

        acesso à  variável  global e quando estivermos dentro da procedure,

        teremos acesso à variável local.

┌─────────────────┐

│VIII - Functions.│

└─────────────────┘

VIII.1 - Definiçäo

         ---------

        As funçöes säo muito parecidas com as procedures. A principal dife-

        rença é que o identificador de uma funçäo assume o valor de retorno

        da funçäo.  Uma  funçäo  deve  sempre  retornar um valor e em Turbo

        Pascal, este valor é retornado no nome da funçäo.

VIII.2 - Declaraçäo de funçöes.

         ----------------------

        A  declaraçäo  de  uma funçäo é muito parecida com de uma procedure

        que por sua vez é parecida com a de um programa, senäo vejamos:

        ---------------------------------------------------------

        Function Nome_da_funçäo(parâmetros) :  Tipo_da_funçäo;

        < área de declaraçöes >

        Begin

                corpo da funçäo

        End;

        ---------------------------------------------------------

        A  formaçäo  do  nome  da funçäo deve seguir as mesmas regras  para

        formaçäo de identificadores em Turbo Pascal. Dentro  dos parenteses

        devemos declarar os parâmetros e seus respectivos tipos dos quais a

        funçäo depende. O tipo de valor retornado pela funçäo  também  deve

        ser declarado.

        Na  área  de   declaraçöes,  podemos  declarar  labels, constantes,

        variáveis e até mesmo Procedures e Functions. Devemos  lembrar  que

        tais elementos só poderäo ser utilizados dentro do corpo da funçäo,

        pois säo locais a ela. Abaixo, temos o exemplo de uma funçäo.

        ---------------------------------------------------------

        Program Exemplo;

        Uses CRT;

        Var x,y : Real;                   (* variáveis globais *)

        Function Soma(a,b:real):real;     (* Soma é uma funçåo

                                             que depende de dois

                                             parâmetros reais e

                                             devolve um valor

                                             real *)

        Begin

           Soma:=a+b;                     (* reparem que o valor

                                             da funçåo é retor-

                                             nado p. seu nome *)

        End;

        Begin

           ClrScr;

           x:=Soma(4,5);

           y:=Soma(3,6)-Soma(45.5,5.6);

           Writeln(x:10:2,y:10:2);

           Writeln;

           Write('Valor de x --> ');

           Readln(x);

           Write('Valor de y --> ');

           Readln(y);

           Writeln;

           Writeln(Soma(x,y):10:2);

        End.

        ---------------------------------------------------------

        Devemos  lembrar  que  o Turbo Pascal  possui inúmeras  funçöes  de

        procedures pré-definidas, que iremos ver no decorrer do curso.

        Exemplos:

        ---------------------------------------------------------

        Program Fat;

        Uses CRT;

        (********************************************************

         Programa para calcular o fatorial de um número lido do

         teclado, usando o conceito de Function

        ********************************************************)

        Label inicio,fim;

        Var n : Integer;

            tecla : char;

        Function Fatorial(numero:integer) : Real;

        Var i   : Integer;

            Fat : Real;

        Begin (* da funçåo Fatorial *)

           Fat:=1;

           If numero>1

              Then Begin

                      i:=1;

                      Repeat

                         i:=i+1;

                         Fat:=Fat*i;

                      Until i=numero;

                   End;

           Fatorial:=Fat;

        End; (* da funçäo fatorial *)

        Begin  (* do programa *)

           ClrScr;

        inicio:

           Write('Valor de n (menor que 0 = fim) --> ');

           Readln(n);

           Writeln;

           If n<0

              Then Begin

                      Write('Nao existe fatorial de numeros negativos');

                      Goto fim;

                   End

              Else Writeln('Fatorial de n = ',fatorial(n):10:0);

           Writeln;

           Goto inicio;

        Fim:

        End. (* do programa *)

        ---------------------------------------------------------

        Program Fibonacci;

        Uses CRT;

        (********************************************************

         Programa para determinar um determinado elemento da se-

         quência de Fibonacci. A sequência de fibonacci é defi-

         nida como

                        Fib(0)  =  0

                  Fib(1)  =  1

                  Fib(n)  =  Fib(n-1) + Fib(n-2)

                        Como podemos ver, o elemento atual é determi-

        nado pela soma dos dois elementos anteriores

        ********************************************************)

        Label inicio;

        Var numero:integer;

            tecla : char;

        Function Fib(n:integer):integer;

        Var  a1,a2,i,pe : Integer;

        Begin

           if n=0

              Then Fib:=0

              Else If n=1

                      Then Fib:=1

                      Else Begin

                              a1:=0;

                              a2:=1;

                              i:=1;

                              Repeat

                                 pe:=a1+a2;

                                 i:=i+1;

                                 a1:=a2;

                                 a2:=pe;

                              Until i=n;

                              Fib:=a2;

                           End;

        End;

        Begin

           ClrScr;

        inicio:

           Write('Fib(');

           Read(numero);

           Writeln(') = ',fib(numero));

           Writeln;

           Write('Deseja continuar ? --> ');

           Readln(tecla);

           writeln;

           writeln;

           If tecla='s' Then goto inicio;

        End.

        ---------------------------------------------------------

VIII.3 - Recursividade

         -------------

         A   linguagem Pascal  e  o Turbo Pascal  permitem a utilizaçäo  de

         funçöes recursivas. Uma funçäo é dita recursiva quando ela chama a

         si  mesma. Devemos tomar cuidado ao lidar com esse tipo de funçäo,

         pois  podemos criar loops infinitos. Existem pessoas que têm faci-

         lidade para  pensar  recursivamente  e outras näo. A recursividade

         permite  criar  funçöes elegantes e torna os programas mais fáceis

         de serem entendidos. Abaixo, temos os mesmos programas anteriores,

         só que utilizando o conceito de recursividade.

        ---------------------------------------------------------

        Program Fatorial;

        Uses CRT;

        Label inicio,fim;

        Var n : Integer;

            tecla : char;

        Function Fat(n:integer):real;

        Begin

           if n=0

              Then Fat:=1

              Else Fat:=n*Fat(n-1);   (* repare que estamos

                                         chamando novamente a

                                         funçåo Fat *)

        End;

        Begin

           ClrScr;

        inicio:

           Write('Valor de n (menor que 0 = fim) --> ');

           Readln(n);

           Writeln;

           If n<0

              Then Begin

                      Write('Nao existe fatorial de numeros negativos');

                      Goto fim;

                   End

              Else Writeln('Fatorial de n = ',fat(n):10:0);

           Writeln;

           Goto inicio;

        Fim:

        End.

        ---------------------------------------------------------

        Program Fibonacci;

        Uses CRT;

        Label inicio;

        Var numero:integer;

            tecla : char;

        Function Fib(n:integer):integer;

        Begin

           If n=0

           Then Fib:=0

           Else If n=1

                   Then Fib:=1

                   Else Fib:=Fib(n-1)+fib(n-2);

        End;

        Begin

           ClrScr;

        inicio:

           Write('Fib(');

           Read(numero);

           Writeln(') = ',fib(numero));

           Writeln;

           Write('Deseja continuar ? --> ');

           Readln(tecla);

           writeln;

           writeln;

           If tecla='s' Then goto inicio;

        End.

        ---------------------------------------------------------

